
ЗАГАЛЬНІ ПОНЯТТЯ
«Чому геометрію часто називають холодною і сухою? Однією з причин є те, що вона не здатна описати форму хмари, гори, дерева чи берега моря.  Хмари — це не сфери, гори — не конуси, берегові лінії — не кола, і кора дерев не гладенька, і блискавка не розповсюджується по прямій.»

Б.Мандельброт

Фрактал можна визначити як об’єкт складної форми, який можна отримати у результаті виконання простого ітераційного циклу. Цей термін у 1975 році ввів французький математик Бенуа Мандельброт.

Ітераційність, рекурсивність спричиняють такі властивості фракталів, як самоподібність: окремі частини схожі за формою на весь фрактал в цілому. Латинське слово fractus означає “зроблений з фрагментів”.

У 1977 році Бенуа Мандельброт видав книжку “Фрактальна геометрія природи” (The Fractal Geometry of Nature). У його роботах були використані наукові результати інших вчених, які працювали над цією ж темою ще у 1875-1925 роках, таких як Кантор, Хаусдорф, Жюліа, Пуанкаре, Фату. З тих пір і аж до сьогодні слово “фрактал” залишається модним як серед математиків, так і серед програмістів.

Сам Мандельброт давав таке означення фрактала:

Фракталом називається множина елементів метричного простору, розмірність Хаусдорфа якої більша за топологічну розмірність.

Пояснимо це означення простішою мовою.

Перш за все множина, яка складає фрактал повинна бути метричною. Це означає, що на ній задане поняття величини елемента, його відносного положення у множині.

Топологічна розмірність — поняття, інтуїтивно знайоме більшості людей. Для прикладу, розмірність множини точок гладкої лінії — одиниця; точок гладкої поверхні — двійка; точок гладкого тіла — три. В якості математичної довідки: розмірність зліченної множини рівна нулю, а порожньої множини — мінус одиниці.

Вочевидь, топологічна розмірність є цілим числом. І звичайно, настав момент, коли математиків зацікавило питання про існування множин з дробовою розмірністю. Тоді Хаусдорф узагальнив поняття топологічної розмірності. Він дав визначення розмірності множини, згідно до якого виявилося, що вона не обов’язково є цілим числом.

Тоді й почалися пошуки множин з дробовою розмірністю. І ці пошуки дали свої результати:

1. Знайшлися “ніби-то лінії”, розмірність яких більша за 1, але менша за 2 — це сніжинка Коха, множина Мандельброта тощо.

2. Знайшлися множини розмірністю менше 1, але більше за 0. Це наприклад пил Кантора та йому подібні множини.

3. Знайшлися “ніби-то плоскі фігури”, розмірність яких менше двох. Це килим Серпинського та інші йому подібні.

4. Знайшлися “ніби-то об’ємні тіла”, розмірність яких менша за 3. Це губка Менгера та інші близькі до неї.

5. Врешті, знайшли і криву лінію, розмірність якої точно дорівнює двом! Знамениті криві Пеано та Гілберта — ці криві можуть заповнити квадрат повністю, тобто вони проходять по усіх точках квадрата і ніколи не перетинають самих себе.

6. У природі теж знайшли фрактали. Виявляється, що траєкторія броунівської часточки є фракталом з розмірністю, більшою за 1! 

Що ж дозволяє фракталам мати дробову розмірність? Перш за все справа у тому, що фрактали зовсім не гладкі. Математично їх характеризують як у повній мірі розривні. Саме ця “супер-розривність” дозволяє їм збільшувати чи зменшувати розмірність. Окрім того фрактали є нескінченно самоподібними. Тобто весь фрактал за формою схожий на як завгодно малу його частину.
За принципом побудови прийнято розрізняти алгебраїчні, геометричні та стохастичні фрактали. Побудова алгебраїчних фракталів базується на аналізі точок площини (простору) і присвоєнні їм певного кольору за якоюсь ознакою. Геометричні фрактали являють собою рекурентні геометричні побудови. Стохастичні фрактали отримуються у тому випадку, коли у циклічному процесі випадковим чином змінюються якісь їх параметри. При цьому можна отримати реалістичні ландшафти, хмари, несиметричні дерева тощо.

Алгебраїчні фрактали
Найпростіший принцип побудови мають алгебраїчні фрактали. Але не зважаючи на це вони мають значно складнішу візуальну структуру, ніж, наприклад, геометричні.

Алгебраїчні фрактали  вперше запропонував Бенуа Мандельброт, якого по праву вважають батьком фрактальної графіки. Він спробував рекурсивно перетворювати точки комплексної площини.

Мандельброт перетворював комплексне число за формулою:

 — початкова точка комплексної площини.

Кожна точка комплексної площини після певної кількості таких перетворень приходить у один з можливих стійких станів (або, як ще кажуть, аттракторів).

Таким чином, весь фазовий простір комплексної площини розбивається на області притягання аттракторів. Тоді, розфарбувавши області притягання різними кольорами, можна отримати так званий фазовий портрет ітераційного процесу.

Несподівано для математиків це призвело до можливості породження складних образів нетривіальної структури за допомогою простих ітераційних процесів.

Для побудови фрактала Мандельброта (див. мал. 1) достатньо для кожної точки z=x+i*y комплексної площини за описаною вище формулою проводити ітерації, поки ця точка не вийде за межі кола з радіусом 2 (тобто доки модуль числа z не перевищуватиме 2). Таким чином фрактал Мандельброта повністю міститься у діапазоні –2<x<2, –2<y<2.

Колір точки зображення фрактала при побудові можна визначати кількістю проведених над нею рекурентних перетворень (ця кількість визначатиме область притягання, у яку потрапила точка).

Можливий випадок, коли точка при перетвореннях достатньо довго не виходить за межі кола з радіусом 2. Тоді можна зробити цю точку чорною (або фіолетовою, як на малюнку).

Взагалі-то, Мандельброт спочатку не мав на увазі такий малюнок. Він досліджував ті точки комплексної площини, які при описаних перетвореннях ніколи не виходять за межі кола радіусом 2 з центром у початку координат.

На межі аттракторів утворюються найскладніші і найцікавіші візерунки, завдяки яким алгебраїчні фрактали так цікавлять тисячі розумів по всьому світові.

Алгебраїчні, або, як їх ще називають, комплексні фрактали — дуже варіативна річ. Достатньо у формулі перетворення комплексного числа змінити один символ — і ви одержите зовсім інший фрактал.

Ось, наприклад, множина Жюліа. Вона має формулу 

, де 

 є деякою комплексною константою. Дуже схоже на формулу Мандельброта, чи не так? Проте малюнок, який одержується, аж ніяк не схожий на фрактал Мандельброта. Окрім того, навіть у одній цій формулі, змінивши константу 

 ми одержимо різні зображення.

На базі перетворень, аналогічних до тих, які проводив Бенуа Мандельброт, за минулі роки були виведені сотні ітераційних формул, які утворюють феєричні картини.

Навіть початківець може намалювати власний алгебраїчний фрактал. Для цього достатньо замість запропонованих вище формул придумати власні формули.
А ось наглядний приклад програми, яка будує алгебраїчні фрактали. У ній, окрім дієвої частини, реалізовано основні операції над комплексними числами: знаходження модуля, піднесення до квадрату, множення, ділення, інкремент (додавання). Ці операції можуть знадобитися читачеві для побудови власних формул перетворення, отримання власних фрактальних зображень.
{$N+}

uses crt,graph;

const x1 = -2.2; x2 = +1.0;

      y1 = -1.2; y2 = +1.2;

      epsx = (x2-x1)/1024;
      epsy = (y2-y1)/768;

      gm1024x768 = 3;

type complex = record
       x,y:extended;
     end;

var GD,GM,c,i:integer;

    z,z0:complex;

function c_abs(c:complex):extended;

begin
 c_abs:=sqrt(sqr(c.x)+sqr(c.y));

end;

procedure c_sqr(var z:complex);

var t:complex;

begin
 t.x:=z.x*z.x-z.y*z.y;

 t.y:=2*z.x*z.y;       z:=t;

end;

procedure c_mul(var z:complex;
                    d:complex);

var t:complex;

begin
  t.x:=z.x*d.x-z.y*d.y;

  t.y:=z.x*d.y+z.y*d.x; z:=t;

end;

procedure c_div(var z:complex;
                    d:complex);

var t:extended;

begin
 if c_abs(d)>0.0 then
  begin
   t:=sqr(d.x)+sqr(d.y);
   d.y:=-d.y;  c_mul(z,d);

   z.x:=z.x/t; z.y:=z.y/t;

  end;

end;

procedure c_inc(var z:complex;
                    d:complex);

begin
 z.x:=z.x+d.x;

 z.y:=z.y+d.y;

end;

procedure pixel(z:complex;c:byte);

begin
 PutPixel(round((z.x-x1)*epsx),
          round((z.y-y1)*epsy),c);

end;

begin
 GD:=InstallUserDriver('svga256',nil);

 GM:=gm1024x768;
 InitGraph (GD,GM,'');

 for i:=0 to 63 do
     setRGBpalette (i,i,i,i);

 z0.x:=x1;
 repeat
  z0.y:=y1;

  repeat
   z:=z0; c:=0;

   while (c<64)and(c_abs(z)<=2.0)
   do begin
        c_sqr(z);  c_inc(z,z0);
        Inc (c);

      end;

   if c<64 then pixel(z0,c);

   z0.y:=z0.y+epsy;

  until z0.y>y2;
  z0.x:=z0.x+epsx;

 until (z0.x>x2) or keypressed;

 readkey;

 CloseGraph;

end.

Вашим додатковим завданням залишається динамічне масштабування малюнку (за допомогою клавіатури або миші), вибір палітри кольорів тощо.

Мабуть ви помітили, що для ініціалізації графіки використано нестандартний драйвер – SVGA256. Він підтримує 256 кольорів і роздільну здатність до 1024 х 768 пікселів. Цей драйвер можна знайти у багатьох місцях мережі Інтернет, зокрема на сайті http://www.all4pupil.iatp.org.ua. Взагалі-то він не є обов’язковим, але погодьтеся: 256-кольорові малюнки з високою роздільною здатністю виглядають значно привабливіше, ніж 16-кольорові з порівняно низькою (640 х 480 пікселів).
Геометричні фрактали
Геометричні фрактали названі так, оскільки їх форма може бути описана, як послідовність простих геометричних операцій (малювання геометричних фігур). Ці фрактали можуть описати деякі природні об’єкти: дерева, рослини, ландшафти тощо. Окрім того, за їх допомогою можна описати різноманітні геометричні об’єкти складної природи.
Існує два основних шляхи будування геометричних фракталів: метод L-систем та метод IFS-систем.

L-системи

Назва способу L-систем пішла від прізвища її винахідника — Лінденмейера.

L-система – це синтаксис деякої простої мови, яка дозволяє виконувати малювання геометричних фігур через команди так званому виконавцю. Часто для пояснення цих систем проводять аналогії з широко відомою навчальною мовою „Лого”.

У мові „Лого” був виконавець – черепашка, – якому можна було давати команди типу “повернися ліворуч”, “повернися праворуч”, “зроби крок уперед” тощо. Відповідно до команди виконавець змінював свій напрямок або координати.

Для малювання деяких геометричних фракталів цілком достатньо трьох операцій:

· розвернутися на заданий кут;

· пройти на заданий крок вперед;

· пройти на заданий крок, залишаючи за собою слід.

Але геометричні фрактали можуть мати складну структуру. Тоді необхідно ускладнити систему команд.

При класичному підході вводять два поняття: аксіома та теорема.

Аксіомою назвемо послідовність команд, яку виконує “черепашка”. У цій послідовності введемо наступні позначення:

· + поворот за годинниковою стрілкою на фіксований кут α;
· – поворот проти годинникової стрілки на той же кут α;

· f крок вперед без малювання;
· F крок вперед змалюванням;
· [ запам’ятати координати і напрямок;

· ] повернутися у останню запам’ятовану позицію (змінити координати та напрямок);

· велика латинська літера — виконати відповідну теорему.

Кожну теорему позначатимемо великою латинською літерою. Теорема — це фактично рекурсивна аксіома. У її записі можуть міститися усі ті символи, які можуть міститися у аксіомі.  Єдина різниця між цими двома поняттями у тому, що при виклику теореми до неї передається степінь рекурсивної заглибленості. Ця заглибленість, по-перше, впливає на масштаб зображення, що малюється; по-друге, за домовленістю, якщо теорема має назву F, то при досягненні бажаної степені деталізації замість виконання теореми відбувається малювання лінії (аналог команди “пройти на заданий крок, залишаючи за собою слід”).

Слід зауважити, що метод L-систем будує передфрактал заданого порядку, тобто вигляд фракталу із заданою скінченною точністю (Раніше було сказано, що фрактали нескінченно самоподібні, але комп’ютер не має поняття „нескінченність”).
Реалізація методу L-систем не є складною, проте вона потребує уваги і розуміння написаного.

uses crt,graph;

type TPosition = record
     X,Y,Angle,Step:real;

end;

type TStack = object
 Content:array [1..2000]

            of TPosition;

 Top:TPosition;

 Last:byte;

 procedure Push (X,Y,ang,st:real);

 procedure Pop;

end;

procedure TStack.Push
                (X,Y,ang,st:real);

begin
  Inc (Last);

  Content[Last].X:=X;

  Content[Last].Y:=Y;

  Content[Last].Angle:=Angle;

  Content[Last].Step:=Step;

  Top.X:=X; Top.Y:=Y;

  Top.Angle:=Angle;

  Top.Step:=Step;

end;

procedure TStack.Pop;

begin
 Content[Last].X:=0;

 Content[Last].Y:=0;
 Content[Last].Angle:=0;

 Content[Last].Step:=0;

 if Last=1 then
   FillChar (Top,SizeOf (Top),0)

 else
 begin
  Top.X:=Content[Last-1].X;

  Top.Y:=Content[Last-1].Y;

  Top.Angle:=Content[Last-1].Angle;

  Top.Step:=Content[Last-1].Step;

 end;

 Dec (Last);

end;

var
  GD,GM:integer;

  Axiom:string;

  Theorem:array ['A'..'Z']
                 of string;

  X,Y,Angle,StAngle:real;

  L,k:real;

  Stage,i,j:byte;

  Stack:TStack;

  Xs,Ys,Ls:real;

procedure Fractal (Theor:char;
                   Stage:byte);

var i:integer;

begin
 MoveTo (round (X),round (Y));

 if (Stage=0)and(Theor='F') then
  begin
   X:=X+L*cos(StAngle*Pi/180);

   Y:=Y+L*sin(StAngle*Pi/180);

   LineTo (round (X),round (Y));

  end else
 if Stage<>0 then
  begin
   for i:=1 to Length
              (Theorem[Theor]) do
    if keypressed then halt else
    case Theorem[Theor][i] of
     'A'..'Z':
      begin
       L:=L*k;

       Fractal (Theorem[Theor][i],
                          Stage-1);

       L:=L/k;

      end;

     'f':
      begin
       X:=X+L*cos(StAngle*Pi/180);

       Y:=Y+L*sin(StAngle*Pi/180);

       MoveTo(round (X),round (Y));

      end;

     '+': StAngle:=StAngle+Angle;

     '-': StAngle:=StAngle-Angle;

     '[': Stack.Push (X,Y,StAngle,L);

     ']':

      begin
       X:=Stack.Top.X;
       Y:=Stack.Top.Y;
       StAngle:=Stack.Top.Angle;

       L:=Stack.Top.Step;

       Stack.Pop;

      end;

    end;

  end;

end;

begin
 InitGraph (GD,GM,'');

 Axiom:='F';

 Theorem['F']:='F-F++F-F';

 Angle:=60; Stage:=7; k:=1/3;

 Xs:=40; Ys:=GetMaxY*3 div 4;
 Ls:=530;

 for i:=1 to Stage do
   begin
    X:=Xs; Y:=Ys;
    StAngle:=0; L:=Ls;

    for j:=1 to Length (Axiom) do
     case Axiom[j] of
      'A'..'Z':

       Fractal (Axiom[j],i);

      'f':
       begin
        X:=X+L*cos(StAngle*Pi/180);

        Y:=Y+L*sin(StAngle*Pi/180);

        MoveTo(round (X),round (Y));

       end;

      '+': StAngle:=StAngle+Angle;

      '-': StAngle:=StAngle-Angle;

      '[': Stack.Push(X,Y,StAngle,L);

      ']':

       begin
        X:=Stack.Top.X;
        Y:=Stack.Top.Y;
        StAngle:=Stack.Top.Angle;

        L:=Stack.Top.Step;

        Stack.Pop;

       end;

     end;

    Readkey;

    ClearDevice;

   end;

 CloseGraph;

end.
Нижче наведено кілька класичних прикладів геометричних фракталів, побудованих за допомогою L-систем.
[image: image34.png]


Крива Коха є одним із найтиповіших прикладів геометричних фракталів. Вона була винайдена у XIX столітті німецьким математиком на ім’я Хельге фон Кох, який при вивченні робіт Георга Кантора та Карла Вейєрштрасса наштовхнувся на опис кривих з незвичайними властивостями.

Аксіома: F

Теорема: F => F-F++F-F

Кут повороту: 60
Мандельброт багато експериментував з кривими Коха, і одержав так звані Острови Коха, Хрести Коха, Сніжинки Коха та інші.

Хрест Коха утворюється, якщо замість аксіоми намалювати квадрат чи прямокутник. Сніжинка Коха утворюється за аксіомою F++F++F.

Окрім того, Бенуа Мандельброт винайшов об’ємне представлення кривої Коха — це тетраедр, у центрах граней якого знаходяться інші тетраедри з меншою стороною.

Фракталом Мандельброта часто називають множину Мандельброта (алгебраїчний фрактал), проте існує і геометричний фрактал, придуманий Мандельбротом — це ще один з варіантів кривої Коха.

[image: image35.png]


Крива Мінковського, або, як її іноді називають, ковбаса Мінковського, утворюється за нескладною формулою:

Аксіома: F

Теорема: F => F-F+F+FF-F-F+F

Кут повороту: 90

За що вона отримала назву ковбаси, вам може відповісти достовірно лише той, хто так її назвав уперше. Дехто знаходить у цій кривій певні асоціації, але ковбаса могла прийти на думку лише дуже голодній людині (.

Крива дракона, або “помах дракона”, була винайдена італійським математиком Джузеппе Пеано. Вона дещо схожа на криву Мінковського. Мандельброт назвав цей фрактал „рікою подвійного дракона”.

Аксіома: X

Теореми: X => X+YF+

[image: image36.png]


         Y => -FX-Y

Кут повороту: 90

Знамениті криві Гілберта і Пеано проявляють ще незвичнішу властивість. Вони обидві не перетинають самі себе і можуть заповнити квадрат повністю, якщо запустити нескінченний цикл ітерацій. Таким чином лінія, яка за звичайних умов одновимірна (не має навіть товщини!) у цих кривих має площу квадрата!
Аксіома: X

Гілберт: X => -YF+XFX+FY-

         Y => +XF-YFY-FX+

Пеано:   X => XFYFX+F+YFXFY-F-XFYFX

         Y => YFXFY-F-XFYFX+F+YFXFY

Кут повороту: 90

[image: image37.png]



IFS-системи

Метод “систем ітеративних функцій” (Iterated Functions Systems, IFS) з’явився у середині 80-х років минулого сторіччя як засіб отримання фрактальних структур.

IFS являє собою систему функцій, які відображають одну багатовимірну множину на іншу. Найпростіший варіант IFS — афінні перетворення на площині (зсув, масштабування, обертання):

Окрім того, для побудови IFS застосовуються і інші класи перетворень, наприклад проективні:

або квадратичні:

Розглянемо приклад IFS-системи для побудови тієї ж кривої Коха, яку ми будували за допомогою L-системи.

Ця крива має чотири частини, які схожі на усю криву. Для її побудови необхідні чотири афінних перетворення координат. Нехай нам задані координати точок A та B. Обчислимо за ними координати точок C, D та E.

Точка C ділить відрізок AB у відношенні 1:2; точка D ділить цей же відрізок у відношенні 2:1; а точка E є точкою D, повернутою на 60 градусів відносно точки C. З цих міркувань випливають наступні формули:

E: 

Тут 

 — координати точки A, 

, 
[image: image15.wmf]2

y

 — координати точки B.

Тепер для утворення кривої Коха достатньо повторити описані афінні перетворення для відрізків AC, CE, ED та DB.

uses crt,graph;

const rmin  = 1;

      alpha =  2*Pi/180;

      beta  = 86*Pi/180;

      k = 0.14; k1 = 0.3;

var GD,GM,i:integer;

procedure step(x1,y1,x2,y2:real;
                   num:integer);

var x3,y3,x4,y4,x5,y5:real;

begin
 if sqr(x1-x2)+sqr(y1-y2)>
               sqr(rmin) then
  begin
   x3:=(2*x1+x2)/3;

   y3:=(2*y1+y2)/3;

   x4:=(x1+2*x2)/3;

   y4:=(y1+2*y2)/3;

   x5:=x3+(x4-x3)/2+(y4-y3)*0.866;

   y5:=y3-(x4-x3)*0.866+(y4-y3)/2;

   step(x1,y1,x3,y3,num+1);

   step(x3,y3,x5,y5,num+1);

   step(x5,y5,x4,y4,num+1);

   step(x4,y4,x2,y2,num+1);

  end else
  begin
   SetColor(num);

   Line(round(x1),round(y1),
        round(x2),round(y2));
  end;

end;

begin
 GD:=InstallUserDriver('svga256',nil);
 GM:=3;

 InitGraph (GD,GM,'');

 for i:=1 to 15 do
    setRGBpalette (i,i*8,i*8,i*8);

 step(getmaxx/4,getmaxy/2,
      getmaxx*3/4,getmaxy/2,1);

 readkey;

 CloseGraph;

end.

[image: image39.png]


Крива дракона Хартера-Хейтвея будується ще простіше. Вона має лише дві самоподібні частини. Тому, маючи координати відрізка AB, можна знайти координати точки C (відрізок AC у 
[image: image16.wmf]2

 менший за відрізок AB, і повернутий відносно точки A на кут 45 градусів):

Далі необхідно зробити ті ж перетворення з відрізками CA та CB. Зверніть увагу, що їх необхідно виконати для відрізка CA, а не для відрізка AC. Подумайте, чому (подивіться уважно на малюнок другого кроку фракталу дракона)?

procedure step(x1,y1,x2,y2:real;
                   num:integer);

var x3,y3,x4,y4,x5,y5:real;

begin
 if sqr(x1-x2)+sqr(y1-y2)>
               sqr(rmin) then
  begin
   x3:=x1+(x2-x1)/2-(y2-y1)/2;

   y3:=y1+(x2-x1)/2+(y2-y1)/2;

   step(x3,y3,x1,y1,num+1);

   step(x3,y3,x2,y2,num+1);
  end else
  begin
   SetColor(num);

   Line(round(x1),round(y1),
        round(x2),round(y2));
  end;

end;

Метод IFS-систем, як бачимо, дає нам простіший та потужніший інструмент для побудови фракталів, ніж метод L-систем. Так, він легко дозволяє виводити рекурентні формули для фракталів (що ж до попереднього методу, то для виведення формул складних фракталів необхідно мати неабияке абстрактне мислення). Ось, наприклад, закон побудови листка папороті:
Точка 3: 

 відносно точки 1.
Точка 4: 

 – поділ відрізка 1-3 у відношенні k.
Точка 5: 

 – поділ відрізка 3-4 у відношенні 

Точка 6: 

 – поворот точки 5 на кут 

 відносно точки 4.

Точка 7: 

 відносно точки 4.
procedure step(x1,y1,x2,y2:real;
                   num:integer);

var x3,y3,x4,y4,x5,y5,

    x6,y6,x7,y7:real;

begin
 if sqr(x1-x2)+sqr(y1-y2)>
               sqr(rmin) then
  begin
   x3:=x1+(x2-x1)*cos(alpha)
         -(y2-y1)*sin(alpha);

   y3:=y1+(x2-x1)*sin(alpha)
         +(y2-y1)*cos(alpha);

   x4:=x1*(1-k)+x3*k;

   y4:=y1*(1-k)+y3*k;

   x5:=x4*(1-k1)+x3*k1;

   y5:=y4*(1-k1)+y3*k1;

   x6:=x4+(x5-x4)*cos(beta)
         -(y5-y4)*sin(beta);

   y6:=y4+(x5-x4)*sin(beta)
         +(y5-y4)*cos(beta);

   x7:=x4+(x5-x4)*cos(-beta)
         -(y5-y4)*sin(-beta);

   y7:=y4+(x5-x4)*sin(-beta)
         +(y5-y4)*cos(-beta);

   setcolor(num);

   Line(round(x1),round(y1),
        round(x4),round(y4));

   step(x4,y4,x3,y3,num);

   step(x4,y4,x6,y6,num+1);

   step(x4,y4,x7,y7,num+1);
  end;
end;

Як бачите, у основу методу систем ітеративних функцій покладено лише кілька нескладних у розумінні принципів. Та й сам алгоритм побудови не відрізняється складністю.
Цікаві ітеративні досліди
Часто виконуючи, здавалося б, випадкові та безсистемні ітераційні перетворення можна отримати дуже складні та красиві зображення. Давайте поекспериментуємо з такими процесами.

Колір кожної точки малюнку 11 отримано за формулою:

Малюнок виглядає, як об’єднання сімей періодичних картинок з колами. Причини такого ефекту і досі залишаються загадкою.
[image: image42.png]


На малюнку 12 зображено лінії рівня власної функції оператора Лапласа. Це функції вигляду:

Вигляд малюнка може змінюватися в залежності від кількості доданків у сумі. Поданий малюнок отриманий при n = 5.
Ще одним цікавим для вивчення об’єктом є так звана послідовність Морса. Вона генерується за таким алгоритмом: спочатку записується 0, далі до вже отриманої послідовності приписуємо її інверсію (замість кожного 0 записуємо 1, а замість кожної 1 записуємо 0). Повторюючи описані дії багато разів одержимо послідовність: 011010011001011010010110...
Ви спитаєте: чим же ця послідовність цікава? З точки зору теорії послідовностей її визначною властивістю є те, що жодна її підпослідовність не повторюється більше двох разів підряд. А з точки зору теорії хаосу та фрактальної графіки вона дає алгоритм побудови дуже складних кривих.
Давайте згадаємо метод L-систем для побудови фрактальних зображень. Команди, які ми кодували символами „+”, „-”, „F”, „f” можна кодувати і по-іншому – наприклад, двійковими кодами. Згрупуємо цифри послідовності Морса по три – утвориться послідовність, елементами якої є 6 різних комбінацій нулів та одиниць (тобто усі можливі комбінації, окрім 000 та 111, які не будуть присутні у послідовності, виходячи зі сформульованої вище властивості цієї послідовності).
Тепер можна кожній такій комбінації поставити у відповідність поворот на певний кут. Наприклад,

	Послідовність
	001
	010
	011
	100
	101
	110

	Кут повороту
	+45
	+45
	+135
	+135
	–90
	+45

Ідучи по послідовності Морса і щоразу рухаючись на один піксел уперед, можна отримати малюнок, схожий на мал. 13. Для цього слід збільшувати яскравість піксела, на який ми попадаємо у процесі побудови.
[image: image45.jpg]


Цікаво, що за допомогою іншого кодування поворотів можна з послідовності Морса отримати криву Коха (див. мал. 14). Це таке кодування:
	Послідовність
	001
	010
	011
	100
	101
	110

	Кут повороту
	–135
	–135
	+90
	–135
	+90
	+90

Ще одним різновидом рекурентних кривих є криві, зображені на мал. 15-16. Хочете – вірте, хочете – ні, але вони побудовані за одними і тими ж формулами. Для їх отримання достатньо взяти x та y рівними 0, а далі перетворювати ці координати точки зображення за формулою:

[image: image30.wmf].

Тут a, b та c – деякі константи. Саме від них залежить, яке зображення ви отримаєте.

Прикладна важливість фракталів
Де ж використовують фрактали у нашому житті? Чи це не просто красиві малюнки, чи містять вони якийсь сенс, користь? Спробуємо відповісти на ці запитання, з’ясуємо аспекти використання фракталів у нашому житті.
Фрактальне стиснення зображень
За допомогою фракталів можна економно зберігати і швидко обробляти зображення з порівняно невеликою втратою якості зображення. Фрактальне стискання інформації дає кращі результати, ніж стиснення JPEG, і методи основані RLE, LZW або кодах Хаффмена. Перелічені методи стиснення не враховують природи даних, які підлягають стисненню, тому дають незадовільні результати при обробці зображень. При фрактальному стисненні зображення конвертується у формат IFS, у матриці афінних перетворень, тому метод є набагато економнішим, аніж просто BITMAP або LZW.

Стиснення JPEG переводить зображення у амплітудно-фазову форму. При цьому області високих частот частково відкидають, оскільки вони не сильно впливають на якість зображення. Саме через це цей метод погано зберігає зображення з тонкими лініями: тексти або креслення.

Що ж до фрактального стиснення, то основною його проблемою є те, що компресія — декомпресія відбуваються швидко, а пряма зміна зображення вимагає великих комп’ютерних ресурсів. Знаходження правильного набору фракталів для кодування зображення є складною задачею науки про розпізнавання образів.

Стиснення відбувається так. Зображення розбивається на частини множиною R та покривається множиною D. При цьому елементи множини D більші за площею, ніж R. Для кожного елементу R перебираються елементи D і будується афінне перетворення з R у D. З усіх перетворень обирається одне, яке має найменшу похибку. Афінне перетворення при цьому відбувається за трьома параметрами: x- ,y-координата та колір точки зображення:

a, b, c, d, e, f - коефіцієнти перетворення на площині;

s - контрастність;

o - яскравість.

Множина таких перетворень і складає систему IFS.

Розбиття на R і покриття D є найскладнішою частиною алгоритму. У найпростішому випадку зображення розбивають сіткою 8*8 пікселів. Далі для кожного елемента сітки перебирають усі можливі елементи D, які є ділянками по  16*16 (тут можливі 8 випадків порівняння: повороти на 90 градусів та симетрії). Для кожного елемента сітки вводять його найкраще перетворення, а потім шукають мінімальне покриття зображення з обраних елементів D і залишають лише ті перетворення, які необхідні для мінімального покриття.

Декомпресія відбувається так. Обирають початкове зображення, з якого буде побудовано оригінал. Найдивовижнішим фактом є те, що від вибору початкового зображення залежить лише час декомпресії, але результуюче зображення буде завжди однаковим! Після обирання зображення до кожного елемента множини R (тобто ділянки розміром 8*8 з цього зображення) застосовують випадково вибране афінне перетворення з тих, що є у побудованій на компресії IFS. В результаті елементи зображення копіюються у інші частини екрану (кінцевого зображення) зі зміною орієнтації, яскравості та контрасту. Коли перебрані усі елементи R, процес повторюється знову.

Поступово зображення буде стабілізуватися, так що після 10-20 ітерацій ми отримаємо зображення, яке було закодовано за допомогою IFS.

При компресії можна навіть не зберігати оригінальні розміри зображення, достатньо запам’ятати їх співвідношення. Тоді при декомпресії можна задавати довільний розмір зображення, масштабуючи його відповідним чином. Якщо обробляється природний об’єкт (наприклад, граніт, пейзаж тощо), то при збільшенні зображення відносно початкових розмірів деякі частини його заміщуватимуться подібними, то ж втрати якості не будуть помітні оком взагалі.
Фрактали у композиції
Молодий голландський музикознавець Нільс Путтеманс називав чотири основних способи використання фракталів у музиці:

1. Конвертація графічного зображення. Саме зображення фрактала конвертується у двовимірну партитуру (координати висота звуку — час), при чому може враховуватися як графічне розташування точок, так і їх колір.

2. За формулою: визначення висот та довжин звуків за допомогою арифметичних розрахунків згідно фрактальних алгоритмів.

3. За генезисом. Певний музичний параметр трансформується таким чином, що результат цього заповнення демонструє фрактальний характер (тобто фактично фрактали не використовуються, але якби результат композиції треба було закодувати формулою, то для цього б найпригіднішими були фрактали).

4. Фрактальна композиція через музичну поведінку. Алеаторні структури, у яких виконавці взаємодіють на основі певних правил, створюючи під час п’єси ніби модель поведінки у складних системах.

Більшість віддають перевагу одній-двом з цих категорій. У США це покоління учнів Беббіта, такі композитори, як Ч.Додж, Г.Нельсон, М.МакНабб, Б.Еванс, Л.Остин, Ч.Вуорінен. У Європі це перш за все Д.Лігетті та Д.К.Літтл.

Конвертація графіки

На практиці цей метод дає найменш цікаві результати, адже  у музиці не такою очевидною є гармонія пропорцій. Існує багато програм, які переводять фрактали у звуки, одна з найвідоміших — «Хорошо темперированный фрактал» Роберта Грінхауза з Каліфорнії. У ній десять різних типів фракталів можуть бути озвучені за допомогою 21-го варіанта звукоряду (цікаво, що у них відсутня хроматична гама). Звуковий результат може варіюватися динамічно, за артикуляцією, октавним розташуванням. Він записується у стандартний MIDI-файл. Автор програми не ставив перед собою задачі отримати цілі твори; він пропонував класти результати програми за основу майбутніх композицій.

Для прикладу, на мал. 17 ви бачите фрактал Іванова, за яким була створена п’єса самого Р.Грінхауза. Композитор розвинув її і придумав додаткові партії для ударних інструментів.

За формулою

Розглянемо твір американського композитора Гаррі Лі Нельсона під назвою “Мандрівка галери Йота” (Voyage of Golah Iota). П’єса відображує мандрівку від Середземномор’я до Південної Америки на міфічному кораблі під назвою Йота.

В основі композиції лежить “сценарій переходу до хаосу” бельгійського математика Ферхюльста (40-ві роки XIX століття) - це був аналіз приросту популяції комах. При цьому, якщо коефіцієнт приросту менше 2.5, то відбувається швидка стабілізація навколо одного аттрактора, далі біфуркації (розмноження) навколо двох, чотирьох, восьми аттракторів, аж поки при збільшенні коефіцієнта приросту не наступає хаос, у якому тим не менше спостерігається спрямування до найменших аттракторів. Цей графік є само подібним при наближенні. Він був використаний у п’єсі Нельсона на двох рівнях:

1. Як огинаюча всього твору: розподіл кількості подій (значення коефіцієнту приросту збільшується до 4 у точці “золотого перерізу” п’єси, а потім зменшується до нуля).

2. Для вироблення подій у мікротоновому звукоряді: значення функції розподілені у діапазоні семи октав (192 тони у октаві). Тут не тільки мікротонова, але й мікроритмічна техніка: техніка звукових мікрочасточок, – «гранул», які звучать менше 50 мсек. Висока щільність «гранул» (до кількох тисяч за секунду), створена за допомогою електронної апаратури, породжує надзвичайний ефект: багатоскладовий звуковий комплекс з багатим, динамічно змінним в часі спектром, не зв’язаним напряму зі звучанням якоїсь однієї певної «гранули».

Не дивлячись на те, що поданий метод не є «синтезом звуку» у традиційному сенсі, він, тим не менше, носить назву «гранулярного синтезу», оскільки звукові події, отримані цим методом, як правило, неможливо синтезувати ні одним з відомих нині методів, окрім того додавання «гранул» відбувається на настільки дрібному рівні, що це ніяк не можна віднести до якогось ритму.

За генезисом

Розглянемо два приклади застосування фракталів у цьому аспекті.

Згаданий вище Чарльз Додж – американський композитор, учень Люенінга та Беббіта у Пристонському університеті, професор Бруклінського та Дартмутського коледжу, співавтор широковідомої книжки "Комп’ютерна музика: синтез, композиція та виконання". Цікаве у його композиції “Profile” композиції: синтез мовних та вокальних звуків, нерідко із застосуванням елементів серійної техніки.

П’єса має триголосний поліфонічний склад. Кожна лінія мелодії отримана стохастичним методом (генератором фрактального шуму). Генерація нот завершується після появи деякої кількості звукових класів (наприклад, для верхнього голосу їх три); тепер навпроти кожної ноти верхнього голосу будується фраза у середньому голосі - така ж генерація, до п’яти звукових класів; потім проти кожної ноти у другому голосі - фраза (до чотирьох класів) у нижньому голосі. В результаті отримується свого роду триголосний нестрогий мензуральний канон (голоси не ідентичні, але самоподібні - як різні рівні у фракталах).

Звуковий результат — відчуття загального взаємозв’язку та статики (також завдяки використанню інтерпольованих синтезованих тембрів).

Другий приклад фрактального твору — фортепіаний етюд Д.Лігетті «L'escalier du diable» («Чортові сходи») – і номер його обраний не випадково: 13.

Лігетті ніколи не використовував у своїй музиці фрактальні формули напряму, віддаючи перевагу їх загальноасоціанивній дії. “Прямо перенесення положень фрактальної геометрії я не вважаю серйозним підходом до створення музики. Натомість я намагаюся знайти музичні аналогії фрактальним фігурам, але без калькулятора, без математики”. Подібні аналогії присутні, наприклад, в IV частині його фортепіанного концерту – мелодичні моделі, які побудовані у ітеровану систему, тобто систему зі зворотнім зв’язком. Музика починається дуже тонко, прозоро, окремими конструкціями. Поступово вона ущільнюється за допомогою самонанизування, накладання цих конструкцій самих на себе.

Назва вищезгаданого етюду має пряме відношення до математики: “чортові східці” – феномен, пов’язаний з так званою множиною Кантора: вона має нерівні “сходинки”, майже усюди вертикальну крутизну, і уходить у нескінченність. Лігетті будує свої «музичні східці», втіливши відношення «двоїстості» та «троїстості» через по-особливому організоване чергування ритмічних комірок – 2/8 и 3/8. Початок п’єси будується на метричній моделі, що складається з “підгруп”: 7, 9, 11 і 9 восьмих (після півторатактової «ввідної побудови»). Звуковисотна модель одразу виступає у асинхронії з ритмічною: кожна нова «комірка» іде з наступної хроматичної гамми; своєрідну гармонію будують сходинки цілотонового звукоряду, в результаті чого отримується «диявольське» поєднання цілотонових звуків з хроматикою. Отримується ніби нескінченний підйом, неможливість досягнення вершини, «шлях у нікуди».

За поведінкою

Іноді композитори у своїх творах створюють своєрідну «партитуру» та звід правил для виконавців. Зокрема, у творі Д.К.Літтла “Мозкова хвиля” (Brain Wave) троє музикантів імпровізують не тільки згідно заданим правилам, але й у зв’язку з діями інших музикантів, які сидять попереду і позаду них (тобто деяке колективне несвідоме).

Позитивний вплив

Якщо музиканти попереду мене грають, то я можу:

· почати грати;

· грати голосніше:

· більшу кількість разів програти частину п’єси;

· намагатися імітувати манеру гри інших музикантів;

· намагатися грати з такою ж швидкістю, як інші.

Негативний вплив

Якщо музиканти позаду мене грають, то я можу:

· припинити грати;

· грати м’якіше, тихіше;

· зупинитися або ж перейти до наступної частини п’єси;

· намагатися грати у манері, протилежній до інших музикантів;

· грати швидко, коли інші грають повільно і навпаки.

Композитор тут намагається створити інтерактивну ситуацію не тільки як модель природних процесів, а і як аналогію з роботою людського мозку.

«... музичний експеримент дає конкретні приклади того,
що є і що не є музикою, зіштовхуючи їх у одному творі.
Ці експерименти не тільки музичні, але й музично-логічні,

не тільки акти художньої творчості,

але й акти абстрактної і навіть гранично абстрактної думки,
прикладені до музичного та квазі-музичного матеріалу»
(Михайлов А.В. Концепція витвору мистецтва у Теодора В. Адорно.
Теодор В. Адорно. Вибране. Соціологія музики. - М.-СПб., 1999, с.356) 

Інформаційні джерела

Інтернет-ресурси:
1. http://www.fractals.nsu.ru
2. http://www.ghcube.com
3. http://timara.con.oberlin.edu/~gnelson
4. http://fract.narod.ru
5. http://www.codenet.ru/progr/fract
мал. 1. Фрактал Мандельброта





мал. 2. Фрактал Мандельброта. Збільшення у 200 разів





мал. 3. Вигляд фракталу Жуліа при різних комплексних константах.





мал. 6. Крива Гілберта





мал. 5. а) крива Мінковського; б) „Ріка Подвійного Дракона”.





мал. 4. а) крива Коха; б) хрест Коха; в) фрактал Мандельброта.





мал.7. Фрактали Серпінського а) трикутник; б) піраміда; в) губка.





мал. 8. Побудова кривої Коха методом IFS





мал. 9. Побудова дракона Хартера-Хейтвея методом IFS.

